
This document contains information to aid analysts in the application of a spreadsheet program to perform 
a hydrographic separation (classification). This hydrographic classification is a preprocessing step to the 
application of the HEFR program. The classification methods in the spreadsheet assign a code to each day 
in the record to classify each day as an extreme low flow, base flow, high flow pulse, small flood or large 
flood.  There are two methods included in the spreadsheet.  The first, IHA-EFC, implements the method 
used in the IHA software program. (The Nature Conservancy, 2007). The second method, MBFIT, is a 
modification of the Baseflow Index (BFI) approach developed by the Institute of Hydrology (Institute of 
Hydrology, 1980a and 1980b).  The primary output from this spreadsheet is the column labeled Code3 and 
is calculated for both IHA-EFC and MBFIT.  The values from Code3, for the method determined by the 
analyst as the most appropriate, should be copied and pasted into column C of the IHA EFC output sheet 
“daily efcs” in Excel.  Other IHA output sheets in Excel (if present) should be deleted.  Column D of “daily 
efcs” must also be updated; this is most easily accomplished using nested IF statements to associate the text 
values in column D with the numeric codes in column C.  The HERF program should then be executed at 
that time. 
 
Conventions 
 
To save computer time and space the results of many of the calculations in this spreadsheet have been 
converted to values using the Copy and PasteSpecial->values option in Excel.  These values will need to be 
recalculated when updating for a new gauge or when the calibration parameters have been modified. The 
macro UpdateCalcs can be used to update these calculations. It takes a minute or two run. 
 
Yellow cells are parameters that the analyst can modify to calibrate the classification. 
Grey cells have formulas which should not be edited.   
Turquoise cells should be edited by the analyst but are used for display and don’t effect the primary 
calculations. 
 
Steps to update for another site. (Save often) 
 

1. Make a working copy of the spreadsheet and save a clean version. 
2. Determine number of rows needed for period of record being analyzed. Enter the starting and 

ending dates that will be analyzed in cells A3 (start date) and B3 (end date) to have the 
program calculate the last row in the daily (C3) and annual (D3) arrays. 

3. Size the spreadsheet as needed. If the new site contains fewer rows than the Evadale site 
(Evadale is the example site used for the HEFR documentation), delete all rows below the last 
row.  If the new site contains more rows than the Evadale site, insert rows so that the last row 
is what is determined in cell C3.  Be sure to insert rows between existing rows rather than 
dragging the formula down from the existing last row. If rows are properly inserted, all of the 
formulas and charts will update correctly, if not formulas and chart titles to be corrected 
manually. Similarly, cells will need to be inserted or deleted from columns AF and AG to 
properly size the columns that contain annual maximums. 

4. Copy date and flow data (presumably downloaded from the USGS) into cells A43:B43. 
5. Update the gage name in cell A1. 
6. Update Calculations using the UpdateCals macro. 
7. View Graphs – There are three graphs to the far right of the worksheet, scales will likely need 

adjusting.  Graph 1 shows the 25th, 50th, and 75th low flow (i.e., base flow) percentiles for the 
two alternative classification approaches. Graph 2 shows the low/high classification (code1) 
from the IHA-HEFR method. Graph 3 shows the low/high classification (code1) from the 
MBFIT method.  UpdateXaxis macro can be used to view the results in Graphs 2 and 3 
stepping through time. 

 
Macros 
 
There are two macros included in the file used to update calculations and review outputs.   



 
UpdateCals macro is used after making changes to the various base flow classification parameters (cells 
colored Yellow). The macro copies down formulas from AG3 to the row number calculated in cell D3, 
C45:S45 to the row number calculated in cell C3 and from D10:I10 to Row 21. It is only necessary to save 
computer time and space.  The cells in Grey contain formulas that should not be edited. 
 
UpdateXaxis macro is intended to aid in reviewing the results in Graphs 2 and 3.  It is used to step through 
time to view the results of the separation. There are several options (cells colored  Turquoise) in cells Q32, 
Q33, R32, R34 that give the user the ability set the period of record to display. 
 
 
Algorithms 
 
The following sections describe some of the more complex calculations in the spreadsheet, specifically the 
calculation of Code0 for the IHA-EFC algorithm and the Code1 for MBFIT.  Code3 (column H for the 
IHA-EFC method and column O for the MBFIT method) is the final baseflow code that will be used in 
subsequent HERF steps.  Once the user is satisfied with the parameterization of one of these methods, the 
Code3 values can be copied into the code column in the “daily efcs” page. 
 
Code 0 – (IHA-EFC only) 
Implementation of the algorithm described in the IHA users manual (p 15 Version 7 March 2007) to 
classify flow as either (-1) low flow, (-2) ascending limb high flow or (-3) descending limb high flow.  
 
Code0=IF(C43=-1,IF(B44<=$G$33,-1,IF(OR(B44>$G$32,(B44-B43)/B43>$F$26),-2,-1)),IF(C43=-
2,IF(B44<=$G$33,-1,IF((B44-B43)/B43*(-1)>$F$27,-3,-2)),IF(B44<=$G$33,-1,IF(B44>$G$32,-
3,IF((B44-B43)/B43>$F$26,-2,IF(((B44-B43)/B43)*(-1)<$F$27,-1,-3)))))) 
 
If the previous days flow was low flow 
=IF(C43=-1, 
Then if today’s flow is less than the low flow threshold, today is low flow 
 IF(B44<=$G$33,-1, 
Otherwise if today’s flow is greater than the upper threshold OR the rate of change is greater than the rising 
threshold then today is rising otherwise it is still low flow 
  IF(OR(B44>$G$32,(B44-B43)/B43>$F$26),-2,-1)), 
 
If the previous days flow was an ascending limb 
IF(C43=-2, 
Then if today’s flow is less than the low flow threshold, today is low flow 
 IF(B44<=$G$33,-1, 
If the rate of change is greater than the declining rate then today is descending otherwise it is still 
ascending.  (This is a bit counter-intuitive.  If you are on an ascending limb and yesterdays flow was 100 
cfs and today is 99, today is still considered part of the ascending limb since the flow has not started falling 
fast enough to be considered declining) 
  IF((B44-B43)/B43*(-1)>$F$27,-3,-2)), 
 
If the previous days flow was a descending limb 
Then if today’s flow is less than the low flow threshold, today is low flow 
 IF(B44<=$G$33,-1, 
If today’s flow is still greater than the upper threshold it is still descending 
  IF(B44>$G$32,-3, 
If the rate of change is greater than the rising rate then today goes back to rising 
  IF((B44-B43)/B43>$F$26,-2, 
If the rate of change is less than the declining rate then today is low other wise it is still descending 
  IF(((B44-B43)/B43)*(-1)<$F$27,-1,-3)))))) 
 



Code 1  
Classifies each day as (1) low flow or (2) high flow. 
Code 1 calculated for IHA simply so that it can be compared with the MBFIT method. 
Code1 is used to produce the results in Graphs 2 and 3. 
 
Code 1 for the MBFIT option is fairly complex and is discussed in detail below. 
 
First find the local minimum within an N day window.  
This local minimum can be considered the base flow portion of the flow on a given day. 
 
Min=MIN(OFFSET(B44,MAX(-(ROW()-ROW(I$42)-1),-(($K$26-1)/2)),0,MIN(ROW()-
ROW(I$42)+(($K$26-1)/2),$K$26),1)) 
 
This formula is complicated because of the start up days.  The goal is to find the minimum flow in an N day 
window.  So if N = 9, you need to look at the previous 4 days and the subsequent 4 days around the current 
day.  For the first 4 days in the record there are not 4 previous days to access, therefore the code is written 
to only look back as many days as there are (on the third day there would be two previous days) 
 
Second, find turning points.: 
Given three adjacent N-day minimum flows, Q0, Q1, and Q2: 
 
 Q1 is a turning point IF: 
 
 Q1*f<= Q0 and Q1*f <=Q2 
 
where f is a turning-point test factor that must be greater than 0 and should be less than 1. 
 
If percent of the current days flow is less than the local minimum for the flow N days ago and N days hence 
then today is a turning point, (and may  be a base flow day). (Column J) 
 
TP=IF(ROW()-ROW(J$42)-1<$K$26,"",IF(AND($K$27*I44<OFFSET(I44,-
$K$26,0,1,1),$K$27*I44<OFFSET(I44,+$K$26,0,1,1)),I44,"")) 
 
Third determine if the day is base or pulse flow dominated.  
If the current day is a turning point, then determine whether the total flow on that day is base or pulse flow 
dominated.  The test parameter is runoff fraction. If the portion of the flow that is considered runoff or 
pulse flow is greater than the runoff fraction of the base flow portion, then the day is classified as a pulse 
day otherwise it is a base flow day.     
 
For example if the flow on a given day is 110 cfs and the turning point value (= baseflow portion of the 
flow on a given day = minimum within N day window) is 100, should the day be categorized as a baseflow 
day or a pulse flow day?  If the runoff fraction is 0.20 then the runoff fraction times the baseflow portion of 
the flow equals 20, since the pulse flow portion of the flow on this day is 10cfs {abs(baseflow portion-total 
flow)} Perhaps put more simply this days is a baseflow dominated day because the portion of the flow that 
is pulse flow (the 10 cfs) is not a significant enough portion to say the day is pulse flow dominated. 
Increasing the runoff fraction raises the standard on what will be considered a pulse dominated day and will 
result in more days being classified as base flow dominated. 
 
The part of the code that does not include the magnitude thresholds is 
 
IF(J44="",2,IF(ABS(J44-B44)>$K$28*J44,2,1)    (part of the equation in column J) 
 
Forth apply the magnitude thresholds which automatically classify high and low flows. 
The full calculation of Code1 for the modified BFI includes magnitude thresholds (L32 and L33) however 
if user inputs 100 and 0 for the upper and lower percentiles, the program returns the period of record 



maximum and  minimum flow which means they have no effect on the classification, the separation 
becomes a pure Modified BFI without any threshold considerations. 
 
Code1=IF(B44>$L$32,2,IF(B44<$L$33,1,IF(J44="",2,IF(ABS(J44-B44)>$K$28*J44,2,1)))) 
 
More information on the BFI method can be found at http://www.usbr.gov/pmts/hydraulics_lab/twahl/bfi/ 
 
Code 2  
Classifies each day as (0) extreme low flow, (1) low flow, (2) high flow pulse, (3) small flood or (4) large 
flood. 
 
Code 3  
Classifies each day based on the event that it is part of. So although the flow on a particular day may be less 
than the small flood threshold, if that storm event eventually exceeds the small flood threshold, then all the 
days in that event are classified as small flood days. This code is the primary output from the spreadsheet 
and should be copied and pasted into IHA code column on the “daily efcs” sheet in HEFR. 
 
Checks 
 
Can be deleted. 

